Query Answering over Linked Data

Project Report — Semantic Web

Raphaël Binda Mathieu Waharte

May 2026

Abstract

This report explores query answering over Linked Data using SPARQL and federated queries across publicly available knowledge graphs. We first demonstrate the expressive power of the Semantic Web stack by answering a biomedical question (proteins involved in signal transduction and related to pyramidal neurons) that cannot be handled by large language models alone due to their lack of real-time, structured biological data. We then propose and answer a complex geopolitical question about COVID-19 vaccine developers using Wikidata. Finally, we discuss how `CONSTRUCT` queries provide formal justification for Linked Data answers, and how hybrid Linked Data / LLM pipelines combine the complementary strengths of both paradigms.

Contents

1 Introduction

The World Wide Web, originally designed for human consumption, has evolved towards a machine-readable layer known as the *Semantic Web* or *Web of Data* [1]. At its core, the Semantic Web relies on the Resource Description Framework (RDF) to express knowledge as triples (*subject–predicate–object*), and on SPARQL as the standard query language to retrieve and manipulate that knowledge. The Linked Open Data (LOD) cloud [3] today contains billions of such triples distributed across hundreds of interconnected datasets, covering domains from biomedicine (UniProt, Bgee, LinkedLifeData) to culture and encyclopaedic knowledge (DBpedia, Wikidata).

In his 2009 TED talk, Tim Berners-Lee challenged the scientific community to unlock the potential of Linked Data for answering complex, cross-domain questions [2]. One such question — "*What proteins are involved in signal transduction and also related to pyramidal neurons?*" — illustrates precisely why the Semantic Web is indispensable: it requires joining heterogeneous, specialised databases in a single, structured, reproducible query, something that Large Language Models (LLMs) alone cannot reliably do.

This project addresses three tasks:

Q1. Demonstrate a Semantic Web method to answer the above biomedical question using federated SPARQL.

Q2. Propose and answer a new complex question — which organisations developed COVID-19 vaccines, and in which countries are they based? — using Wikidata.

Q3. Explain and justify the answers using `CONSTRUCT` queries and discuss a hybrid Linked Data + LLM approach.

2 Problem Statement

Answering complex factual questions requires three capabilities that simple keyword search or LLMs lack: (i) **access to up-to-date, structured data**, (ii) **the ability to join information across heterogeneous sources**, and (iii) **provenance and justification** of the answer.

LLMs like ChatGPT encode world knowledge statically at training time. They cannot query live databases, cannot guarantee that their answers reflect the current state of specialised repositories (e.g. UniProt protein annotations), and cannot easily provide a formal justification (a *proof*) of how an answer was derived from source data.

The Semantic Web stack — RDF, OWL, SPARQL, and the federated query mechanism introduced in SPARQL 1.1 — directly addresses these limitations. The challenge is to translate natural-language questions into correct SPARQL queries, navigate the data models of multiple endpoints, and combine the results coherently.

3 Question 1 — Biomedical Federated Query

3.1 Question

What proteins are involved in signal transduction and also related to pyramidal neurons?

This question was posed by Tim Berners-Lee in his 2009 TED talk as an example of a query that could, in principle, be answered by connecting biological databases on the Web, but that was not yet tractable at the time.

3.2 Datasets explored

- **UniProt** (`https://sparql.uniprot.org/sparql`): the reference database for protein sequences and functional annotation. Each protein can be annotated with Gene Ontology (GO) terms, including GO:0007165 (*signal transduction*).
- **Bgee** (`https://www.bgee.org/sparql`): a database of gene expression across anatomical entities and developmental stages. It provides expression data for genes in specific cell types such as pyramidal neurons.
- **Gene Ontology (OBO)** (`https://purl.obolibrary.org/obo/`): the controlled vocabulary linking biological processes to GO terms, used to navigate the hierarchy of signal transduction subprocesses.

3.3 Method

The key insight is that the two conditions in the question are held in *different* databases:

1. Signal transduction annotation lives in UniProt (via GO terms).
2. Expression in pyramidal neurons lives in Bgee (via anatomical entity labels).

SPARQL 1.1's `SERVICE` keyword enables federated queries: a single SPARQL query can delegate sub-queries to remote endpoints and join the results locally. The bridge between the two databases is the UniProt identifier, which Bgee exposes through the `lscr:xrefUniprot` predicate.

3.4 SPARQL Query

```
PREFIX up:     <http://purl.uniprot.org/core/>
PREFIX go:     <http://purl.obolibrary.org/obo/>
PREFIX lscr:   <http://purl.org/lscr#>
PREFIX genex:  <http://purl.org/genex#>
PREFIX orth:   <http://purl.org/net/orth#>
PREFIX rdfs:   <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?protein ?proteinName ?goTerm ?goLabel WHERE {

  # Condition 1: Protein involved in signal transduction (
     UniProt)
```

```
  SERVICE <https://sparql.uniprot.org/sparql> {
    ?protein a up:Protein ;
             up:recommendedName/up:fullName ?proteinName ;
             up:classifiedWith ?goTerm .
    ?goTerm rdfs:subClassOf* go:GO_0007165 .  # signal transduction
    ?goTerm rdfs:label ?goLabel .
  }

  # Condition 2: Gene expressed in pyramidal neurons (Bgee)
  SERVICE <https://www.bgee.org/sparql> {
    ?gene a orth:Gene ;
          genex:isExpressedIn ?cond .
    ?cond genex:hasAnatomicalEntity ?anatEntity .
    ?anatEntity rdfs:label "pyramidal neuron"@en .
    ?gene lscr:xrefUniprot ?protein .
  }
}
```

Listing 1: Federated SPARQL query joining UniProt and Bgee.

3.5 Difficulties encountered

- **Data model exploration**: the predicate connecting a protein to an anatomical entity is not obvious. We first had to inspect the Bgee ontology to discover the `genex:isExpressedIn` / `genex:hasAnatomicalEntity` chain.
- **GO hierarchy**: signal transduction has many sub-processes. Using `rdfs:subClassOf*` (transitive closure) correctly captures all relevant GO terms without hardcoding each one.
- **Cross-database identifier alignment**: UniProt protein URIs must match Bgee's `lscr:xrefUniprot` values. This required checking that both sides use the canonical UniProt accession format.
- **Endpoint availability**: federated queries depend on remote endpoints being online and responsive; timeouts can occur for broad queries.

4 Question 2 — COVID-19 Vaccines and Their Developers

4.1 Question

Which COVID-19 vaccines were developed by which organisations, and in which country are those organisations based?

4.2 Datasets explored

Wikidata (https://query.wikidata.org) was chosen as the primary source. Wikidata is a collaboratively maintained knowledge graph that covers a wide range of domains, including public health and pharmacology. Vaccine entries are linked to their developers via the `wdt:P178` (*developer*) property, and developers are linked to their country of headquarters via `wdt:P17` (*country*). The class `wd:Q84263196` identifies COVID-19 vaccines.

4.3 SPARQL Query

```
PREFIX wd:        <http://www.wikidata.org/entity/>
PREFIX wdt:       <http://www.wikidata.org/prop/direct/>
PREFIX wikibase:  <http://wikiba.se/ontology#>
PREFIX bd:        <http://www.bigdata.com/rdf#>

SELECT DISTINCT ?vaccine ?vaccineLabel
                ?developer ?developerLabel
                ?country ?countryLabel
WHERE {
  ?vaccine wdt:P1924 wd:Q84263196 ;   # instance of: COVID-19
     vaccine
           wdt:P178  ?developer .      # developer
  ?developer wdt:P17 ?country .        # country

  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "en" .
  }
}
ORDER BY ?countryLabel ?vaccineLabel
```

Listing 2: Wikidata SPARQL query for COVID-19 vaccine developers.

4.4 Sample results

Vaccine	**Developer**	**Country**
CoronaVac	Sinovac Biotech	China
ArVac Cecilia G.	UNSAM / CONICET	Argentina
COVAX-19	Flinders U. / Vaxine	Australia
Ad26.COV2.S	Janssen Pharmaceutica	Belgium
Comirnaty (BNT162b2)	BioNTech / Pfizer	Germany / USA
Oxford–AstraZeneca	University of Oxford / AZ	United Kingdom
Sputnik V	Gamaleya Institute	Russia

Table 1: A selection of results returned by the Wikidata query.

4.5 Role of LLMs in this task

ChatGPT was used in a complementary way: we asked it to *suggest* which Wikidata properties and item identifiers to use (e.g. `P1924`, `Q84263196`), which significantly reduced the time spent browsing the Wikidata property catalogue. However, the actual, structured, and up-to-date answer was retrieved exclusively from Wikidata through SPARQL — ChatGPT alone would risk returning outdated or fabricated vaccine–country associations.

5 Question 3 — Explanation, Justification, and Hybrid Approach

5.1 Justifying answers with CONSTRUCT

The `SELECT` form of SPARQL returns tabular results. To *explain* an answer — i.e. to expose the exact RDF triples that support it — SPARQL's `CONSTRUCT` form builds an explicit sub-graph. This sub-graph constitutes a machine-readable *proof* or justification.

For Question 2, the following `CONSTRUCT` query returns the RDF graph that justifies each vaccine–developer–country answer:

```
PREFIX wd:        <http://www.wikidata.org/entity/>
PREFIX wdt:       <http://www.wikidata.org/prop/direct/>
PREFIX wikibase:  <http://wikiba.se/ontology#>
PREFIX bd:        <http://www.bigdata.com/rdf#>
PREFIX rdfs:      <http://www.w3.org/2000/01/rdf-schema#>
PREFIX schema:    <http://schema.org/>

CONSTRUCT {
  # The vaccine is an instance of a COVID-19 vaccine
  ?vaccine wdt:P1924 wd:Q84263196 .

  # The vaccine was developed by this organisation
  ?vaccine wdt:P178 ?developer .

  # The organisation is based in this country
  ?developer wdt:P17 ?country .

  # Human-readable labels for traceability
  ?vaccine  rdfs:label ?vaccineLabel .
  ?developer rdfs:label ?developerLabel .
  ?country  rdfs:label ?countryLabel .
}
WHERE {
  ?vaccine wdt:P1924 wd:Q84263196 ;
           wdt:P178  ?developer .
  ?developer wdt:P17 ?country .

  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "en" .
    ?vaccine   rdfs:label ?vaccineLabel .
    ?developer rdfs:label ?developerLabel .
    ?country   rdfs:label ?countryLabel .
  }
}
LIMIT 50
```

Listing 3: `CONSTRUCT` query for Q2 justification.

The resulting RDF graph makes the derivation fully explicit: each answer triple is individually grounded in Wikidata statements, providing both **transparency** (where the data comes from) and **reproducibility** (anyone can re-run the query on the live endpoint).

5.2 How the hybrid Linked Data + LLM approach works

Neither Linked Data nor LLMs are sufficient alone. Table 2 summarises their complementary strengths.

Linked Data (SPARQL)	**LLMs (ChatGPT)**
Structured, machine-readable data	Natural language understanding
Always up-to-date (live endpoints)	Static knowledge (training cut-off)
Formally provable answers	No formal provenance
Requires precise query formulation	Handles vague / ambiguous questions
Difficult for non-expert users	Accessible to everyone
Can join multiple databases	Limited to training data

Table 2: Comparison of Linked Data and LLM capabilities.

A practical hybrid pipeline proceeds as follows:

1. **Intent parsing (LLM)**: the natural language question is given to an LLM, which identifies entities (“COVID-19 vaccine”, “developer”, “country”), relevant ontology terms, and suggests candidate SPARQL triple patterns and property identifiers.

2. **Query generation (LLM + human validation)**: the LLM generates a draft SPARQL query. The user or an automated validator checks that the properties and classes exist in the target knowledge graph.

3. **Execution (Linked Data)**: the validated query is executed against the live SPARQL endpoint (Wikidata, UniProt, Bgee, etc.) and returns structured, up-to-date results.

4. **Answer generation (LLM)**: the tabular results are fed back to the LLM, which summarises them in fluent natural language and highlights notable patterns (e.g. “Most approved vaccines were developed in partnership between a pharmaceutical company and a university”).

5. **Justification (CONSTRUCT)**: optionally, a `CONSTRUCT` query retrieves the supporting RDF sub-graph, which is shown to the user as a provenance certificate.

This pipeline mirrors emerging architectures such as *Retrieval-Augmented Generation* (RAG), applied here to structured knowledge graphs rather than unstructured text corpora.

5.3 Visualization of Q2 results via CONSTRUCT

The following `CONSTRUCT` query builds an RDF graph representation of the vaccine–developer–country relationships for Question 2:

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

CONSTRUCT {
  ?vaccine wdt:P1924 wd:Q84263196 ;
           wdt:P178 ?developer ;
           rdfs:label ?vaccineLabel .

  ?developer wdt:P17 ?country ;
             rdfs:label ?developerLabel .

  ?country rdfs:label ?countryLabel .
}
WHERE {
  ?vaccine wdt:P1924 wd:Q84263196 ;
           wdt:P178 ?developer .
  ?developer wdt:P17 ?country .

  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

Listing 4: `CONSTRUCT` query for Q2 result visualization.

This query returns an explicit RDF graph where the interconnections between vaccines, their developers, and countries are visualized as named triples. Figure 1 shows the resulting graph structure.



Figure 1: RDF graph visualization of vaccine–developer–country relationships returned by the `CONSTRUCT` query in Listing 4.

6 Discussion

6.1 Benefits of Linked Data

- **Federated access**: a single SPARQL query can join data from UniProt, Bgee, and GO without pre-downloading or merging the datasets.
- **Open standards**: RDF and SPARQL are W3C standards, ensuring interoperability and long-term stability.
- **Provenance**: `CONSTRUCT` queries make justification first-class, enabling auditability.
- **Precision**: results are exact matches to the query, not probabilistic generations.

6.2 Limitations of Linked Data

- **Steep learning curve**: formulating correct SPARQL queries requires knowledge of the target ontologies and data models.
- **Endpoint fragility**: federated queries fail if any remote endpoint is unavailable or slow.
- **Data completeness**: Wikidata and other community-driven graphs may have missing or inconsistent entries.
- **No reasoning over free text**: Linked Data cannot interpret documents, images, or unstructured annotations without additional NLP.

6.3 Benefits and limitations of the hybrid approach

The hybrid pipeline inherits the benefits of both paradigms: the accessibility of LLMs and the precision and provenance of Linked Data. Its main limitation is error propagation: an incorrectly generated SPARQL query (hallucinated property names, wrong class URIs) will produce wrong or empty results with no obvious error message for the end user. Human-in-the-loop validation at the query generation step is therefore essential.

7 Conclusion

This project demonstrated that the Semantic Web stack remains uniquely suited for answering complex, cross-domain factual questions that require structured, live, and justified data retrieval. Federated SPARQL queries, spanning UniProt, Bgee, and Wikidata, provide answers that LLMs cannot reliably generate from training data alone. The `CONSTRUCT` mechanism adds formal explainability, a critical feature for scientific and high-stakes applications. Finally, the hybrid Linked Data + LLM pipeline offers a pragmatic path forward: LLMs lower the barrier to query formulation, while Linked Data guarantees accuracy, freshness, and provenance.

References

[1] T. Berners-Lee, J. Hendler, and O. Lassila, "The Semantic Web," *Scientific American*, vol. 284, no. 5, pp. 34–43, 2001.

[2] T. Berners-Lee, "The Next Web" (TED Talk), 2009. https://www.ted.com/talks/tim_berners_lee_the_next_web

[3] M. Schmachtenberg, C. Bizer, and H. Paulheim, "Adoption of the Linked Data Best Practices in Different Topical Domains," in *Proc. ISWC*, 2014. https://lod-cloud.net

[4] S. Harris, A. Seaborne, et al., "SPARQL 1.1 Query Language," W3C Recommendation, March 2013. https://www.w3.org/TR/sparql11-query/

[5] UniProt Consortium, "UniProt: the Universal Protein Knowledgebase in 2023," *Nucleic Acids Research*, vol. 51, D1, pp. D523–D531, 2023. https://sparql.uniprot.org/sparql

[6] F. Bastian et al., "Bgee in 2023: Improved coverage and new features for gene expression data in animals," *Nucleic Acids Research*, vol. 51, D1, pp. D1951–D1958, 2023. https://www.bgee.org/sparql

[7] D. Vrandecic and M. Krötzsch, "Wikidata: A Free Collaborative Knowledgebase," *Communications of the ACM*, vol. 57, no. 10, pp. 78–85, 2014. https://query.wikidata.org

[8] Ontotext, "Linked Life Data," https://linkedlifedata.com, 2023.